Program Example 14  Moving Buttons 2

Objectives

Program Purpose

• To further annoy the user and delight the programmer
• To show simple animation using visible, form width, form height, top and left properties
• To demonstrate the use of Message Box options

Learning Goals

• To further develop the use of simple loops
• To use random numbers
• To alter the form size at runtime
• To use multiple colours and form movement
• To use visible and invisible properties to show and hide objects
• To use a module
• To control start up from a ‘Sub MAIN’

Design Notes

This program is based on the project ‘Moving Buttons 1’, but requires a few important changes. This task introduces:

• A module
• A ‘control array’
• A ‘Sub MAIN’ (the project properties have to be altered so that the program starts with the ‘Sub MAIN’ rather than the form)

Create one command button first, then copy and paste it to produce the others. Answer ‘Yes’ when asked if you want to create a control array. The order in which the buttons are pasted is only important only for the three in the lower-left corner. Give these the index numbers: 18, 0, 19. The other buttons can be assigned the numbers between 0 and 17 in any order.

To add a module to a project, go to PROJECT/ADD MODULE on the menu bar.

Interface

Create the interface as shown below.
Use 20 command buttons and 1 timer.
### Names of Objects

<table>
<thead>
<tr>
<th>Type of Object</th>
<th>Number</th>
<th>Names of Objects</th>
<th>Simple Initial Properties of Objects</th>
</tr>
</thead>
<tbody>
<tr>
<td>Form</td>
<td>1</td>
<td>frmRegistrationNow</td>
<td>Caption – “Moving Buttons 2</td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td>Startup Position - 2 – Center Screen</td>
</tr>
<tr>
<td>Timer</td>
<td>1</td>
<td>tmrCountDown</td>
<td>Font – Bold, 10</td>
</tr>
<tr>
<td>Command Buttons</td>
<td>20</td>
<td>cmdRegister(18) to cmdRegister(19)</td>
<td>Font – Bold, 8: Captions – “Register</td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td>Later”, “Help”</td>
</tr>
</tbody>
</table>

### Further Initial Properties of Objects

<table>
<thead>
<tr>
<th>Object</th>
<th>Property</th>
<th>Initial Value</th>
</tr>
</thead>
<tbody>
<tr>
<td>tmrCountDown</td>
<td>Interval</td>
<td>100</td>
</tr>
<tr>
<td></td>
<td>Enabled</td>
<td>True</td>
</tr>
<tr>
<td>CmdRegister(0) to cmdRegister(17)</td>
<td>Caption</td>
<td>“Register Now”</td>
</tr>
<tr>
<td></td>
<td>Font</td>
<td>Normal</td>
</tr>
<tr>
<td>CmdRegister(0) to cmdRegister(16)</td>
<td>Visible</td>
<td>False</td>
</tr>
<tr>
<td>CmdRegister(17) to cmdRegister(19)</td>
<td>Visible</td>
<td>True</td>
</tr>
</tbody>
</table>

### Events – To place in the module (includes declarations and much of the code)

```vba
Const ButtonWidth = 1000
Const ButtonHeight = 500
Const MouseGap = 200

Public Const RegistrationTimeLimit = 25

Const msgHelp1 = "Welcome to the XYZ Program. Please register now to unlock the 30-day expiry code on this program. If you dont activate registration then the program will end automatically. ")
Const msgHelp2 = " seconds to register!"
Const msgRegister0 = "Congratulations! You outwitted the slow old computer!"
Const msgRegister1 = "Are you sure you want to register later?"
Const msgRegister2 = "Are you really sure you want to register later?"
Const msgRegister3 = "Are you really, really sure you want to register later?"
Const msgRegister4 = "Wrong Choice, Dude!"

Const msgTitle1 = "Think Carefully!"
Const msgTitle2 = "Persistent Aren’t You?"
Const msgTitle3 = "Warning!"
Const msgTitle4 = "No More Chances!"

Const NumButtons = 20

Public TimeGone As Single 'used to keep track of time left to register
Public SecondsLeft As Single 'as above
Public vbResponse As Integer 'used to store user’s responses to Message Box dialog boxes

Public frmRed As Integer
Public frmGreen As Integer
Public frmBlue As Integer

Public GoingUp As Boolean
```
Sub Main()
Dim RandomNum As Integer
Randomize
RandomNum = Int(Rnd() * 3) + 1 'picks a number randomly - 1,2 or 3
frmRegistrationNow.Show
Select Case RandomNum   'randomly does option 1, 2 or 3
Case 1
    MoveFormInCircle
Case 2
    ShrinkForm
Case 3
    ReplicateButtons
End Select
End Sub

Sub delay(X As Single)
'useful for delaying program events
Dim starttime As Single
starttime = Timer
While starttime + X > Timer
    DoEvents
    Wend
End Sub

Sub OpenRegistration()
'Create an actual registration form and open in this event
MsgBox "This opens up a registration form which asks for payment along with all your personal details.", , "Hand over your money!"
End Sub

Sub ShrinkForm()
'form contract settings
Const FormContractSpeed = 0.1
Const FormContractAmount = 200
'hide RegisterLaterButton
frmRegistrationNow.cmdRegister(18).Visible = False
'shrinkwidth
While frmRegistrationNow.Width > 1000 'dont make this value too small
    frmRegistrationNow.Width = frmRegistrationNow.Width - FormContractAmount
    delay FormContractSpeed
    Wend
'shrinkheight
While frmRegistrationNow.Height > 500 'dont make this value too small
    frmRegistrationNow.Height = frmRegistrationNow.Height - FormContractAmount
    delay FormContractSpeed
    Wend
'end program finally
End
Sub AlterColor()

' Increases or decreases the RGB values by 20 each time
' Can’t go over 255 or below 0
' Goes up to 255 then goes down to 0
' Uses the Boolean variable “GoingUp” to determine whether to increment or decrement the RGB values

If GoingUp Then
   If frmRed < 235 Then
      frmRed = frmRed + 20
   ElseIf frmGreen < 235 Then
      frmGreen = frmGreen + 20
   ElseIf frmBlue < 235 Then
      frmBlue = frmBlue + 20
   Else
      GoingUp = False
   End If
Else
   If frmRed > 20 Then
      frmRed = frmRed - 20
   ElseIf frmGreen > 20 Then
      frmGreen = frmGreen - 20
   ElseIf frmBlue > 20 Then
      frmBlue = frmBlue - 20
   Else
      GoingUp = True
   End If
End If

frmRegistrationNow.BackColor = RGB(frmRed, frmGreen, frmBlue)

End Sub

Sub ReplicateButtons()

Dim i As Integer, RandomNum As Integer

' Hides and Unhides the series of command buttons
Randomize
frmRegistrationNow.cmdRegister(18).Caption = "Register Now"

'make them visible
For i = 0 To NumButtons - 1
   frmRegistrationNow.cmdRegister(i).Visible = True
delay 0.1
Next i

' randomly swap captions
While SecondsLeft > 0
   RandomNum = Int(Rnd() * NumButtons) ' randomly creates an integer number between 0 and the number of buttons in the form.
   For i = 0 To NumButtons - 1
      If i <> RandomNum Then
         frmRegistrationNow.cmdRegister(i).Caption = "Register Now"
         frmRegistrationNow.cmdRegister(i).FontBold = False
      Else
         frmRegistrationNow.cmdRegister(i).Caption = "Register Later"
         frmRegistrationNow.cmdRegister(i).FontBold = True
      End If
   Next i
   RandomNum = Int(Rnd() * NumButtons)
   frmRegistrationNow.cmdRegister(RandomNum).Caption = "Help"
   frmRegistrationNow.cmdRegister(RandomNum).FontBold = True
delay 0.75

Wend

End Sub
Sub MoveFormInCircle()

Const DelayFormAmount = 0.01
Const MoveForm = 500

Dim TopDist As Integer
Dim LeftDist As Integer
Dim BotDist As Integer
Dim RightDist As Integer

frmRed = 1
frmBlue = 1
frmGreen = 1
GoingUp = True

TopDist = 100
LeftDist = 100
BotDist = Screen.Height - frmRegistrationNow.Height - MoveForm
RightDist = Screen.Width - frmRegistrationNow.Width - MoveForm

While TopDist < BotDist
    'MoveUp until near top
    While frmRegistrationNow.Top > TopDist
        frmRegistrationNow.Top = frmRegistrationNow.Top - MoveForm
        delay DelayFormAmount
        AlterColor
    Wend

    'Move To Top Left Corner
    While frmRegistrationNow.Left > LeftDist
        frmRegistrationNow.Left = frmRegistrationNow.Left - MoveForm
        delay DelayFormAmount
        AlterColor
    Wend

    'Move to Bottom Left Corner
    While frmRegistrationNow.Top < BotDist
        frmRegistrationNow.Top = frmRegistrationNow.Top + MoveForm
        delay DelayFormAmount
        AlterColor
    Wend

    'Move to Bottom Right Corner
    While frmRegistrationNow.Left < RightDist
        frmRegistrationNow.Left = frmRegistrationNow.Left + MoveForm
        delay DelayFormAmount
        AlterColor
    Wend

    TopDist = TopDist + MoveForm
    LeftDist = LeftDist + MoveForm
    BotDist = BotDist - MoveForm
    RightDist = RightDist - MoveForm
Wend

ShrinkForm

End Sub

Sub CountDown()

'Keeps track of how much time left to register
'The amount added to the variable TimeGone must equal the
'interval setting for the timer in the Properties Box
'Remember that this is milliseconds

TimeGone = TimeGone + 0.1
SecondsLeft = RegistrationTimeLimit - TimeGone
If SecondsLeft >= 0 Then
    frmRegistrationNow.Caption = "Registration Form - " & Format(SecondsLeft, ".##" & " Seconds Left!")
Else
    ProgramEnd
End If

End Sub
Sub ProgramEnd()
'Self-Destruct Sequence
ShrinkForm
MoveFormInCircle
End Sub

Sub RegisterLater()
'Uses Nested IF-THEN-ELSE-END-IF structures to annoy the user
vbResponse = MsgBox(msgRegister0 & msgRegister1, vbYesNo + vbDefaultButton2 + vbQuestion, msgTitle1)
If vbResponse = vbYes Then
    vbResponse = MsgBox(msgRegister2, vbYesNo + vbDefaultButton2 + vbExclamation, msgTitle2)
    If vbResponse = vbYes Then
        vbResponse = MsgBox(msgRegister3, vbYesNo + vbDefaultButton2 + vbCritical, msgTitle3)
        If vbResponse = vbYes Then
            MsgBox msgRegister4, vbInformation, msgTitle4
        ProgramEnd
    Else
        OpenRegistration
    End If
Else
    OpenRegistration
End If
Else
    OpenRegistration
End If
End Sub

Sub Help()
frmRegistrationNow.tmrCountDown.Enabled = False
MsgBox msgHelp1 & SecondsLeft & msgHelp2
frmRegistrationNow.tmrCountDown.Enabled = True
End Sub

Events – To place in the form

Private Sub cmdRegister_Click(Index As Integer)
Select Case cmdRegister(Index).Caption
Case "Help"
    Help
Case "Register Later"
    RegisterLater
Case Else
    OpenRegistration
End Select
End Sub

Private Sub tmrCountDown_Timer()
CountDown
End Sub
Notes

When referencing objects on a form from inside a module, the form name has to be included. For example:

```
If i <> RandomNum Then
    frmRegistrationNow.cmdRegister(i).Caption = "Register Now"
    frmRegistrationNow.cmdRegister(i).FontBold = False
Else
    frmRegistrationNow.cmdRegister(i).Caption = "Register Later"
    frmRegistrationNow.cmdRegister(i).FontBold = True
End If
```

Suggestions for Consolidation and Extension

1. Modify the program so that it moves the form at a slower speed.
2. Change the direction the form moves in.
3. Shrink the form as it moves around the screen.
   
   **Hint:** Experiment with the constant values for form width and height reductions, as well as the delay parameters

Questions

1. What is the difference between a 'Public Statement' and a 'Dim Statement'?  
2. Name the 'Boolean Constant' used in this program. How is it used? 
3. What is RGB? 
4. Nested loops are used in this program. Note their location and explain the function of the procedure they are used in. 
5. Write code to randomly produce any integer number between 1 and 10. 
6. What do the following lines of code do? 
   ```
   frmRed = 1  
   frmBlue = 1  
   frmGreen = 1
   ```
7. Explain the function of the 'AlterColor' procedure in plain English. 
8. Explain the function of the sub 'Replicate Buttons'.
9. How do you make a program start with the 'Sub Main' rather than a form?